

Programming Languages: Imperative Program Construction

Practicals 9: Array Manipulation

Shin-Cheng Mu

Autumn Term, 2024

Typical Array Manipulation

- Given $a : \text{array } [0..10] \text{ of } \text{Int}$, compute $\text{wp} (a[i] := 0) (a[2] \neq 0)$.

Solution:

$$\begin{aligned}
 & \text{wp} (a[i] := 0) (a[2] \neq 0) \\
 & \equiv 0 \leq i < 10 \wedge (a:i \rightarrow 0)[2] \neq 0 \\
 & \equiv \{ \text{function alteration} \} \\
 & \quad 0 \leq i < 10 \wedge (i = 2 \Rightarrow 0 \neq 0) \wedge (i \neq 2 \Rightarrow a[2] \neq 0) \\
 & \equiv \{ 0 \neq 0 \equiv \text{False} \} \\
 & \quad 0 \leq i < 10 \wedge (i = 2 \Rightarrow \text{False}) \wedge (i \neq 2 \Rightarrow a[2] \neq 0) \\
 & \equiv \{ P \Rightarrow \text{False} \equiv \neg P \} \\
 & \quad 0 \leq i < 10 \wedge i \neq 2 \wedge (i \neq 2 \Rightarrow a[2] \neq 0) \\
 & \equiv \{ \text{proposition logic} \} \\
 & \quad 0 \leq i < 10 \wedge i \neq 2 \wedge a[2] \neq 0 .
 \end{aligned}$$

- Given constant $N, Y : \text{Int}$ with $0 \leq N$, and variables $b : \text{array } [0..N] \text{ of } \text{Int}, x, i : \text{Int}$,

- compute $\text{wp} (b[i - 1] := x + 1) \langle \forall j : i \leq j < N : b[j] = Y \rangle$.

Solution:

$$\begin{aligned}
 & \text{wp} (b[i - 1] := x + 1) \langle \forall j : i \leq j < N : b[j] = Y \rangle \\
 & \equiv 0 \leq i - 1 < N \wedge \langle \forall j : i \leq j < N : (b:i - 1 \rightarrow x + 1)[j] = Y \rangle \\
 & \equiv \{ \text{since } i - 1 < j, \text{function alteration} \} \\
 & \quad 1 \leq i \leq N \wedge \langle \forall j : i \leq j < N : b[j] = Y \rangle .
 \end{aligned}$$

- Compute $\text{wp} (b[i - 1] := x + 1; i := i - 1) \langle \forall j : i \leq j < N : b[j] = Y \rangle$.

Solution:

$$\begin{aligned}
 & \text{wp} (b[i - 1] := x + 1; i := i - 1) \langle \forall j : i \leq j < N : b[j] = Y \rangle \\
 & \equiv \text{wp} (b[i - 1] := x + 1) \langle \forall j : i - 1 \leq j < N : b[j] = Y \rangle \\
 & \equiv 0 \leq i - 1 < N \wedge \langle \forall j : i - 1 \leq j < N : (b:i - 1 \rightarrow x + 1)[j] = Y \rangle \\
 & \equiv \{ \text{split off } j = i - 1 \} \\
 & \quad 1 \leq i \leq N \wedge (b:i - 1 \rightarrow x + 1)[i - 1] = Y \wedge \\
 & \quad \langle \forall j : i \leq j < N : (b:i - 1 \rightarrow x + 1)[j] = Y \rangle \\
 & \equiv \{ \text{function alteration} \} \\
 & \quad 1 \leq i \leq N \wedge x + 1 = Y \wedge \langle \forall j : i \leq j < N : b[j] = Y \rangle .
 \end{aligned}$$

3. Derive

```

con N : Int {1 ≤ N}
con F : array [0..N) of Int
var h : array [0..N) of Int
running_sum
{⟨∀k : 0 ≤ k < N : h[k] = ⟨Σi : 0 ≤ i ≤ k : F[i]⟩⟩} .

```

Solution: This problem can be seen as a slightly varied instance of Simple Array Assignment mentioned in the handouts. We could have utilised the results. For practice, however, let's start from the basics.

Let $P n \equiv \langle \forall k : 0 \leq k < n : h[k] = \langle \sum i : 0 \leq i \leq k : F[i] \rangle \rangle$. Conjecture the following skeleton:

```

con N : Int {1 ≤ N}
con F : array [0..N) of Int
var h : array [0..N) of Int
var n : Int
initialise
{P 1}
n := 1
{P n ∧ 1 ≤ n ≤ N, bnd : N - n}
do n ≠ N → step
    n := n + 1
od
{⟨∀k : 0 ≤ k < N : h[k] = ⟨Σi : 0 ≤ i ≤ k : F[i]⟩⟩} .

```

Note that $1 \leq N$, and we decided to start the loop with $n = 1$. The *initialise* statement thus has to be $h[0] := F[0]$. (Proof omitted – do it if it is not yet familiar to you!) The reason we start with $n = 1$ will be evident later.

We conjecture that *step* can be performed by a single array assignment $h[I] := E$. We then have to find I and E such that

$$P n \wedge 1 \leq n < N \Rightarrow (P(n+1))[h \setminus (h:I \rightarrow E)] .$$

Let us inspect $P(n+1)$, assuming $1 \leq n < N$:

$$\begin{aligned} & \langle \forall k : 0 \leq k < n+1 : h[k] = \langle \sum i : 0 \leq i \leq k : F[i] \rangle \rangle \\ & \equiv \{ 1 \leq n < N, \text{split off } k = n \} \\ & \quad \langle \forall k : 0 \leq k < n : h[k] = \langle \sum i : 0 \leq i \leq k : F[i] \rangle \rangle \wedge \\ & \quad h[n] = \langle \sum i : 0 \leq i \leq n : F[i] \rangle . \end{aligned}$$

(One could start with expanding $(P(n+1))[h \setminus (h:I \rightarrow E)]$ directly. I find it easier to take it slower.)

Now consider $(P(n+1))[h \setminus (h:I \rightarrow E)]$, assuming $P n \wedge 1 \leq n < N$

$$\begin{aligned} & \langle \forall k : 0 \leq k < n : (h:I \rightarrow E)[k] = \langle \sum i : 0 \leq i \leq k : F[i] \rangle \rangle \wedge \\ & \quad (h:I \rightarrow E)[n] = \langle \sum i : 0 \leq i \leq n : F[i] \rangle \\ & \equiv \{ 1 \leq n < N, \text{split off } i = n. ((*) \text{ see the Think remark in the end}) \} \\ & \quad \langle \forall k : 0 \leq k < n : (h:I \rightarrow E)[k] = \langle \sum i : 0 \leq i \leq k : F[i] \rangle \rangle \wedge \\ & \quad (h:I \rightarrow E)[n] = \langle \sum i : 0 \leq i \leq n-1 : F[i] \rangle + F[n] \\ & \equiv \{ P n \} \\ & \quad \langle \forall k : 0 \leq k < n : (h:I \rightarrow E)[k] = h[k] \rangle \wedge \\ & \quad (h:I \rightarrow E)[n] = h[n-1] + F[n] \\ & \equiv \{ \text{choose } I = n, E = h[n-1] + F[n] \} \\ & \quad \langle \forall k : 0 \leq k < n : h[k] = h[k] \rangle \wedge \\ & \quad h[n-1] + F[n] = h[n-1] + F[n] \\ & \equiv \text{True} . \end{aligned}$$

```

con N : Int { $1 \leq N$ }
con F : array [0..N) of Int
var h : array [0..N) of Int
var n : Int
h[0] := F[0]
{P 1}
n := 1
{P n  $\wedge 1 \leq n \leq N, bnd : N - n$ }
do n ≠ N → h[n] := h[n - 1] + F[n]
    n := n + 1
od
{ $\langle \forall k : 0 \leq k < N : h[k] = \langle \sum i : 0 \leq i \leq k : F[i] \rangle \rangle$ } .

```

In retrospect, we need $1 \leq n < N$ to guarantee $\text{def}(h[n - 1] + F[n])$ (that is, both array accesses are within bound). Therefore we have to start the loop with $n = 1$. Fortunately we can do so because $1 \leq N$.

Think: in the step labelled (*) above, why could we not do the following instead of the splitting?

$$\begin{aligned} & \dots \wedge (h : I \rightarrow E)[n] = \langle \sum i : 0 \leq i \leq n : F[i] \rangle \\ &= \{P n\} \\ & \dots \wedge (h : I \rightarrow E)[n] = h[n] . \end{aligned}$$

Practice: try solving this problem using Simple Array Assignment.

4. Derive

```

con N : Int { $1 \leq N$ }
var f : array [0..N) of Int
con H : array [0..N) of Int
decompose
{ $\langle \forall k : 0 \leq k < N : H[k] = \langle \sum i : 0 \leq i \leq k : f[i] \rangle \rangle$ } .

```

Solution: Similar to the previous exercise, we let $P n \equiv \langle \forall k : 0 \leq k < n : H[k] = \langle \sum i : 0 \leq i \leq k : f[i] \rangle \rangle$, and conjecture the following skeleton:

```

con N : Int { $1 \leq N$ }
con H : array [0..N) of Int
var f : array [0..N) of Int
var n : Int
f[0] := H[0]
{P 1}
n := 1
{P n  $\wedge 1 \leq n \leq N, bnd : N - n$ }
do n ≠ N → step
    n := n + 1
od
{ $\langle \forall k : 0 \leq k < N : H[k] = \langle \sum i : 0 \leq i \leq k : f[i] \rangle \rangle$ } .

```

Conjecture that *step* can be performed by a single array assignment $f[I] := E$. We then have to find *I* and *E* such that

$$P\ n \wedge 1 \leq n < N \Rightarrow (P\ (n+1))[f \setminus (f : I \rightarrow E)] .$$

Inspect $P\ (n+1)$, assuming $1 \leq n < N$:

$$\begin{aligned} & \langle \forall k : 0 \leq k < n+1 : H[k] = \langle \sum i : 0 \leq i \leq k : f[i] \rangle \rangle \\ & \equiv \{ 1 \leq n < N, \text{split off } k = n \} \\ & \quad \langle \forall k : 0 \leq k < n : H[k] = \langle \sum i : 0 \leq i \leq k : f[i] \rangle \rangle \wedge \\ & \quad H[n] = \langle \sum i : 0 \leq i \leq n : f[i] \rangle . \end{aligned}$$

Now consider $(P\ (n+1))[f \setminus (f : I \rightarrow E)]$, assuming $P\ n \wedge 1 \leq n < N$

$$\begin{aligned} & \langle \forall k : 0 \leq k < n : H[k] = \langle \sum i : 0 \leq i \leq k : (f : I \rightarrow E)[i] \rangle \rangle \wedge \\ & \quad H[n] = \langle \sum i : 0 \leq i \leq n : (f : I \rightarrow E)[i] \rangle \\ & \equiv \{ 1 \leq n < N, \text{split off } i = n \} \\ & \quad \langle \forall k : 0 \leq k < n : H[k] = \langle \sum i : 0 \leq i \leq k : (f : I \rightarrow E)[i] \rangle \rangle \wedge \\ & \quad H[n] = \langle \sum i : 0 \leq i \leq n-1 : (f : I \rightarrow E)[i] \rangle + (f : I \rightarrow E)[n] \\ & \equiv \{ \text{choose } l = n, \text{see below (*)} \} \\ & \quad \langle \forall k : 0 \leq k < n : H[k] = \langle \sum i : 0 \leq i \leq k : f[i] \rangle \rangle \wedge \\ & \quad H[n] = \langle \sum i : 0 \leq i \leq n-1 : f[i] \rangle + E \\ & \equiv \{ P\ n \} \\ & \quad \langle \forall k : 0 \leq k < n : H[k] = \langle \sum i : 0 \leq i \leq k : f[i] \rangle \rangle \wedge \\ & \quad H[n] = H[n-1] + E \\ & \equiv \{ \text{choose } E = H[n] - H[n-1] \} \\ & \quad \langle \forall k : 0 \leq k < n : H[k] = \langle \sum i : 0 \leq i \leq k : f[i] \rangle \rangle \wedge \\ & \quad H[n] = H[n-1] + (H[n] - H[n-1]) \\ & \equiv \{ P\ n \} \\ & \quad \text{True} . \end{aligned}$$

In the step marked (*), since $0 \leq i \leq n-1$, by choosing $l = n$ both occurrences of $(f : I \rightarrow E)[i]$ reduce to $f[i]$. Meanwhile, $(f : I \rightarrow E)[n]$ reduces to E .

The derived program is

```

con N : Int {1 ≤ N}
con H : array [0..N) of Int
var f : array [0..N) of Int
var n : Int
f[0] := H[0]
{P 1}
n := 1
{P n ∧ 1 ≤ n ≤ N, bnd : N - n}
do n ≠ N → f[n] := H[n] - H[n - 1]
      n := n + 1
od
{⟨ ∀k : 0 ≤ k < N : H[k] = ⟨ ∑i : 0 ≤ i ≤ k : f[i] ⟩ ⟩} .

```

Swaps

5. Prove that

$$\begin{aligned} & \{h[0] = 0 \wedge h[1] = 1\} \quad \text{-- hence } h[h[0]] = 0 \\ & \text{swap } h(h[0])(h[1]) \\ & \{h[h[1]] = 1\} \end{aligned}$$

Solution: Assume $h[0] = 0 \wedge h[1] = 1$, we have

$$\begin{aligned} & (h:h[0], h[1] \rightarrow h[h[1]], h[h[0]]) \\ & = (h:0, 1 \rightarrow h[1], h[0]) \\ & = (h:0, 1 \rightarrow 1, 0) . \end{aligned}$$

Therefore, let $h' = (h:h[0], h[1] \rightarrow h[h[1]], h[h[0]])$,

$$\begin{aligned} & \wp(\text{swap } h(h[0])(h[1]))(h[h[1]]) = 1 \\ & \equiv h'[h'[1]] = 1 \\ & \equiv h'[0] = 1 \\ & \equiv 1 = 1 \\ & \equiv \text{True} . \end{aligned}$$

6. Given $h : \text{array}[0..N] \text{ of } A$, prove the rule that when h does not occur free in E and F ,

$$\begin{aligned} & \{\langle \forall i : 0 \leq i < N \wedge i \neq E \wedge i \neq F : h[i] = H i \rangle \wedge h[E] = X \wedge h[F] = Y\} \\ & \text{swap } h E F \\ & \{\langle \forall i : 0 \leq i < N \wedge i \neq E \wedge i \neq F : h[i] = H i \rangle \wedge h[E] = Y \wedge h[F] = X\} . \end{aligned}$$

Notes:

- Recall that E and F are expressions, while X, Y, H are logical variables. It means that, for example, one can conclude immediately $X[z \setminus w] = X$ for $z \neq X$, while to determine whether $E[z \setminus w] = E$ we have to look into $E - E[z \setminus w] = E$ if z does not occur free in E .
- With $h[E] = X$, for example, we implicitly assume that $\text{def}(h[E])$ holds.

Solution: Abbreviate $(h:E, F \rightarrow h[F], h[E])$ to h' . We reason:

$$\begin{aligned} & \wp(\text{swap } h E F)(\langle \forall i : 0 \leq i < N \wedge i \neq E \wedge i \neq F : h[i] = H i \rangle \wedge h[E] = Y \wedge h[F] = X) \\ & \equiv \{\text{definition of } \wp; N, H, X, Y \text{ are logical variables}\} \\ & \langle \forall i : 0 \leq i < N \wedge i \neq (E[h \setminus h']) \wedge i \neq (F[h \setminus h']) : h'[i] = H i \rangle \wedge h'[E[h \setminus h']] = Y \wedge h'[F[h \setminus h']] = X \\ & \equiv \{h \text{ does not occur free in } E \text{ and } F\} \\ & \langle \forall i : 0 \leq i < N \wedge i \neq E \wedge i \neq F : h'[i] = H i \rangle \wedge h'[E] = Y \wedge h'[F] = X \\ & \equiv \{\text{function alteration: } h'[i] = h[i] \text{ for } i \neq E \wedge i \neq F\} \\ & \langle \forall i : 0 \leq i < N \wedge i \neq E \wedge i \neq F : h[i] = H i \rangle \wedge h'[E] = Y \wedge h'[F] = X \\ & \equiv \{\text{function alteration}\} \\ & \langle \forall i : 0 \leq i < N \wedge i \neq E \wedge i \neq F : h[i] = H i \rangle \wedge h[F] = Y \wedge h[E] = X . \end{aligned}$$

7. Derive the following program, where arrays are manipulated only by swapping.

```

con N : Int {0 ≤ N}
var h : array [0..N] of Int
var p : Int
?
{0 ≤ p ≤ N ∧ ⟨ ∀i : 0 ≤ i < p : h[i] ≤ 0 ⟩ ∧ ⟨ ∀i : p ≤ i < N : 0 ≤ h[i] ⟩} .

```

Solution: As the usual practice, we use an up-loop in which n is incremented in the end. Let $P n = \langle \forall i : 0 \leq i < p : h[i] \leq 0 \rangle \wedge \langle \forall i : p \leq i < n : 0 \leq h[i] \rangle$. The plan is:

```

con N : Int {0 ≤ N}
var h : array [0..N) of Int
var p, n : Int
p, n := 0, 0
{0 ≤ p ≤ n ≤ N ∧ P n, bnd : N - n}
do n ≠ N → ...n := n + 1 od
{0 ≤ p ≤ N ∧ P N} .

```

Assuming $0 \leq p \leq n < N$, examine $P(n+1)$:

$$\begin{aligned}
& \langle \forall i : 0 \leq i < p : h[i] \leq 0 \rangle \wedge \langle \forall i : p \leq i < n+1 : 0 \leq h[i] \rangle \\
\equiv & \{ \text{since } 0 \leq n < N, \text{ split off } i = n \} \\
& \langle \forall i : 0 \leq i < p : h[i] \leq 0 \rangle \wedge \langle \forall i : p \leq i < n : 0 \leq h[i] \rangle \wedge 0 \leq h[n] \\
\equiv & P n \wedge 0 \leq h[n] .
\end{aligned}$$

Therefore, if $0 \leq h[n]$ there is nothing more we need to do before $n := n + 1$. We can introduce an **if** and put $n := n + 1$ under a guard $0 \leq h[n]$.

To make the **if** total we consider what to do when $h[n] \leq 0$. In this case we consider two cases.

Case: $p \neq n$. We aim to construct

```

{⟨∀i : 0 ≤ i < p : h[i] ≤ 0⟩ ∧ ⟨∀i : p ≤ i < n : 0 ≤ h[i]⟩ ∧ h[n] ≤ 0 ∧ 0 ≤ p < n < N}
???
{P(n+1) ∧ 0 ≤ p ≤ n+1 ≤ N}
n := n + 1
{P n ∧ 0 ≤ p ≤ n ≤ N}

```

Since $p \neq n$, we can split $i = p$ from $\langle \forall i : p \leq i < n : 0 \leq h[i] \rangle$, resulting in

```

{⟨∀i : 0 ≤ i < p : h[i] ≤ 0⟩ ∧ 0 ≤ h[p] ∧ ⟨∀i : p+1 ≤ i < n : 0 ≤ h[i]⟩ ∧ 0 ≤ p < n < N ∧ h[n] ≤ 0}
swap h p n
{⟨∀i : 0 ≤ i < p : h[i] ≤ 0⟩ ∧ h[p] ≤ 0 ∧ ⟨∀i : p+1 ≤ i < n : 0 ≤ h[i]⟩ ∧ 0 ≤ p < n < N ∧ 0 ≤ h[n]}
p := p + 1
{P(n+1) ∧ 0 ≤ p ≤ n+1 ≤ N}
n := n + 1
{P n ∧ 0 ≤ p ≤ n ≤ N} .

```

Case: $p = n$. In this case the range $p \leq i < n$ is empty and the precondition reduces as such:

```

{⟨∀i : 0 ≤ i < p : h[i] ≤ 0⟩ ∧ 0 ≤ p = n < N ∧ h[n] ≤ 0}
???
{P(n+1) ∧ 0 ≤ p ≤ n+1 ≤ N}
n := n + 1
{P n ∧ 0 ≤ p ≤ n ≤ N} .

```

It turns out that the same code still works:

```

{⟨∀i : 0 ≤ i < p : h[i] ≤ 0⟩ ∧ 0 ≤ p = n < N ∧ h[n] ≤ 0}
swap h p n
p := p + 1
{P(n+1) ∧ 0 ≤ p ≤ n+1 ≤ N}
n := n + 1
{P n ∧ 0 ≤ p ≤ n ≤ N} .

```

Therefore the code is:

```

con N : Int { $0 \leq N$ }
var h : array [0..N) of Int
var p, n : Int
p, n := 0, 0
{ $0 \leq p \leq n \leq N \wedge P(n, bnd : N - n)$ }
do  $n \neq N \rightarrow$ 
  if  $0 \leq h[n] \rightarrow n := n + 1$ 
  |  $h[n] \leq 0 \rightarrow swap\ h\ p\ n$ 
    p, n := p + 1, n + 1
  fi
od
{ $0 \leq p \leq N \wedge P(N)$ } .

```

8. The following is a specification of sorting:

```

con N : Int { $0 \leq N$ }
var h : array [0..N) of Int
sort
{ $\langle \forall i j : 0 \leq i \leq j < N : h[i] \leq h[j] \rangle$ } .

```

where *sort* mutates the array *h* only by swapping. Derive a $O(N^2)$ algorithm for sorting. The algorithm will contain a loop within a loop. The outer loop uses as invariant $P_0 \wedge P_1$, where

$$P_0 \equiv \langle \forall i : 0 \leq i < n : \langle \forall j : i \leq j < N : h[i] \leq h[j] \rangle \rangle , \\ P_1 \equiv 0 \leq n \leq N .$$

The inner loop uses *Q* as *part of* its invariant:

$$Q \equiv \langle \forall j : k \leq j < N : h[n] \leq h[j] \rangle .$$

Solution: The invariant is designed such that $n := 0$ establishes $P_0 \wedge P_1$, while $P_0 \wedge P_1 \wedge n = N$ meets the postcondition. Therefore, the outline of the program could be:

```

con N : Int { $0 \leq N$ }
var h : array [0..N) of Int
var n : Int
n := 0
{ $P_0 \wedge P_1, bnd : N - n$ }
do  $n \neq N \rightarrow$ 
  inner_loop
  { $P_0 \wedge P_1 \wedge \langle \forall j : n \leq j < N : h[n] \leq h[j] \rangle \wedge n \neq N$ } -- (*)
  n := n + 1
od
{ $\langle \forall i j : 0 \leq i \leq j < N : h[i] \leq h[j] \rangle$ } .

```

The assertion (*) before $n := n + 1$ is calculated by:

$$\begin{aligned}
& (P_0 \wedge P_1)[n \setminus n+1] \\
& \equiv \langle \forall i : 0 \leq i < n+1 : \langle \forall j : i \leq j < N : h[i] \leq h[j] \rangle \rangle \wedge 0 \leq n+1 \leq N \\
& \Leftarrow \{ \text{with } 0 \leq n < N, \text{ split off } i = n \} \\
& \quad \langle \forall i : 0 \leq i < n : \langle \forall j : i \leq j < N : h[i] \leq h[j] \rangle \rangle \wedge \\
& \quad \langle \forall j : n \leq j < N : h[n] \leq h[j] \rangle \wedge 0 \leq n < N \\
& \equiv \{ \text{def. of } P_0 \text{ and } P_1 \} \\
& \quad P_0 \wedge P_1 \wedge \langle \forall j : n \leq j < N : h[n] \leq h[j] \rangle \wedge n \neq N .
\end{aligned}$$

We now try to construct the *inner-loop*. Compare the hint Q and the assertion (*), we note that

- $P_0 \wedge P_1 \wedge Q \wedge k = n$ establishes (*), and
- letting $k := N - 1$ establishes Q , and
- being in the outer loop, we have $P_1 \wedge n \neq N$, which is $0 \leq n < N$, therefore by choosing $k := N - 1$ we still have $0 \leq n \leq k < N$.

Therefore we start with trying:

$$\begin{aligned}
& \{P_0 \wedge P_1 \wedge n \neq N\} \\
& k := N - 1 \\
& \{P_0 \wedge Q \wedge 0 \leq n \leq k < N\} \\
& \textbf{do } k \neq n \rightarrow \\
& \quad ? \\
& \quad k := k - 1 \\
& \textbf{od} \\
& \{P_0 \wedge P_1 \wedge \langle \forall j : n \leq j < N : h[n] \leq h[j] \rangle \wedge n \neq N\} \\
& n := n + 1
\end{aligned}$$

To construct ? we examine $Q[k \setminus k-1]$, assuming $0 \leq n < k < N$:

$$\begin{aligned}
& \langle \forall j : k \leq j < N : h[n] \leq h[j] \rangle [k \setminus k-1] \\
& \equiv \langle \forall j : k-1 \leq j < N : h[n] \leq h[j] \rangle \\
& \equiv \{ \text{with } 0 \leq n < k < N, \text{ split off } j = k-1 \} \\
& \quad \langle \forall j : k \leq j < N : h[n] \leq h[j] \rangle \wedge h[n] \leq h[k-1] \\
& \equiv Q \wedge h[n] \leq h[k-1] .
\end{aligned}$$

If $h[n] \leq h[k-1]$ already holds, we need only a *skip*. If $h[n] \geq h[k-1]$ holds instead, we do a *swap h n (k-1)*, whose validity can be established by:

$$\begin{aligned}
& \text{wp}(\text{swap h n (k-1)})((P_0 \wedge Q \wedge 0 \leq n \leq k < N)[k \setminus k-1]) \\
& \equiv \{ \text{calculation above, } k \text{ not occurring free in } P_0 \} \\
& \quad \text{wp}(\text{swap h n (k-1)})(P_0 \wedge Q \wedge h[n] \leq h[k-1] \wedge 0 \leq n \leq k-1 < N) \\
& \equiv \{ \text{let } h' = (h : n, k-1 \mapsto h[k-1], h[n]) \} \\
& \quad P_0[h \setminus h'] \wedge Q[h \setminus h'] \wedge h'[n] \leq h'[k-1] \wedge 0 \leq n \leq k-1 < N .
\end{aligned}$$

Consider the first three terms, $h'[n] \leq h'[k-1]$ equivs $h[k-1] \leq h[n]$ by the definition of function alteration (this is why we do *swap h n (k-1)* in the first place). For the second term, we have $Q[h \setminus h'] \Leftarrow Q \wedge h[k-1] \leq h[n]$:

$$\begin{aligned}
& Q[h \setminus h'] \\
& \equiv \langle \forall j : k \leq j < N : h'[n] \leq h'[j] \rangle \\
& \equiv \{ h' = (h : n, k-1 \mapsto h[k-1], h[n]) \text{ and thus } h' j = h j \text{ for } k \leq j < N \} \\
& \quad \langle \forall j : k \leq j < N : h[k-1] \leq h[j] \rangle \\
& \Leftarrow \langle \forall j : k \leq j < N : h[n] \leq h[j] \rangle \wedge h[k-1] \leq h[n] \\
& \equiv Q \wedge h[k-1] \leq h[n] .
\end{aligned}$$

Consider $P_0[h \setminus h']$, assuming $0 \leq n < k < N$:

$$\begin{aligned}
P_0[h \setminus h'] &\equiv \langle \forall i : 0 \leq i < n : \langle \forall j : i \leq j < N : h'[i] \leq h'[j] \rangle \rangle \\
&\equiv \{ h' = (h : n, k - 1 \rightarrow h[k - 1], h[n]) \text{ and } n < k, \text{ thus } h' \text{ } i = h \text{ } i \text{ for } 0 \leq i < n \} \\
&\quad \langle \forall i : 0 \leq i < n : \langle \forall j : i \leq j < N : h[i] \leq h'[j] \rangle \rangle \\
&\equiv \{ \text{with } 0 \leq n < k < N, \text{ split off } j = n \text{ and } j = k - 1 \} \\
&\quad \langle \forall i : 0 \leq i < n : \langle \forall j : i \leq j < N \wedge j \neq n \wedge j \neq k - 1 : h[i] \leq h'[j] \rangle \wedge \\
&\quad \quad h[i] \leq h[n] \wedge h[i] \leq h[k - 1] \rangle \\
&\equiv \{ h' \text{ } j = h \text{ } j \text{ within } i \leq j < N \wedge j \neq n \wedge j \neq k - 1 \} \\
&\quad \langle \forall i : 0 \leq i < n : \langle \forall j : i \leq j < N \wedge j \neq n \wedge j \neq k - 1 : h[i] \leq h[j] \rangle \wedge \\
&\quad \quad h[i] \leq h[k - 1] \wedge h[i] \leq h[n] \rangle \\
&\equiv \{ \text{split off } j = n \text{ and } j = k - 1, \text{ reversed } \} \\
&\quad \langle \forall i : 0 \leq i < n : \langle \forall j : i \leq j < N : h[i] \leq h[j] \rangle \rangle \\
&\equiv P_0 .
\end{aligned}$$

Therefore we have

$$\begin{aligned}
P_0[h \setminus h'] \wedge Q[h \setminus h'] \wedge h'[n] \leq h'[k - 1] \wedge 0 \leq n \leq k - 1 < N \\
\Leftarrow \{ \text{calculation above, some of them assuming } 0 \leq n < k < N \} \\
P_0 \wedge Q \wedge h[k - 1] \leq h[n] \wedge 0 \leq n < k < N .
\end{aligned}$$

Note that, in deriving the inner loop we cannot forget about P_0 — one still has to prove that P_0 is preserved.

In conclusion, the program we derived is:

```

con N : Int {0 ≤ N}
var h : array [0..N) of Int
var n, k : Int
n := 0
{P0 ∧ 0 ≤ n ≤ N, bnd : N - n}
do n ≠ N →
    k := N - 1
    {P0 ∧ Q ∧ 0 ≤ k < N, bnd : k}
    do k ≠ n → if h[n] ≤ h[k - 1] → skip
        | h[n] ≥ h[k - 1] → swap h n (k - 1)
        fi
    k := k - 1
od
{P0 ∧ ⟨ ∀j : n ≤ j < N : h[n] ≤ h[j] ⟩ ∧ 0 ≤ n < N}
n := n + 1
od
{⟨ ∀i j : 0 ≤ i ≤ j < N : h[i] ≤ h[j] ⟩} .

```